

Application Note

Copying CRT Graphics from the HP-83/85 to the 82905B



Hardcopy images of CRT graphics can be generated using the BASIC programs listed below. Binary utility routines also exist for this purpose and are available through the HP Series 80 Users' Library. These utilities will provide a single-command copy capability and will copy much faster than the BASIC routines. Ordering information for these is given at the end of this note.

Program 1—Rotated Copy (3 min.)

Enter the program as shown below and store as "RCOPY"

```
10 !PROGRAM TO COPY HP-83/85 CRT GRAPHICS TO THE 82905B
20 !           ** ROTATED COPY **
30 DIM A$(192),B$(192),G$(7)
40 PRINTER IS 701,80           ! SELECT PRINTER
50 G$=CHR$(27)&"*b192G"       ! GRAPHICS CODES
60 PRINT CHR$(27)&"&19D"      ! SPACING TO 8 DOTS
70 PLOTTER IS 1
80 SCALE 0,255,0,191
90 FOR I=0 TO 31
100 MOVE I*8,191              ! MOVE TO NEXT COLUMN
110 BREAD A$,1                ! READ 1 COLUMN OF THE SCREEN
120 A=POS(A$,CHR$(13))         ! CHECK FOR CR CHARACTER AND REPLACE
130 IF A THEN A$(A,A)=CHR$(12) @ GOTO 100
140 FOR J=1 TO 192
150 B$(193-J,193-J)=A$(J,J)    ! REVERSE ORDER OF BYTES
160 NEXT J
170 SET I/O 7,16,0            ! ELIMINATE END-OF-LINE SEQUENCE
180 PRINT USING 190 ; G$,B$    ! SEND OUT DATA
190 IMAGE 23X,7A,192A
200 SET I/O 7,16,2
210 PRINT USING "K";" "
220 NEXT I
230 PRINT CHR$(27)&"&16D"      ! SPACING TO 6 LPI
240 END
```

Program 2—Horizontal Copy (16 min.)
Enter the program as shown below and store as "HCOPY"

```

10 !PROGRAM TO COPY HP-83/85 CRT GRAPHICS TO THE 82905B
20 !           ** HORIZONTAL COPY **
30 DIM A$(256),G$(7)
40 PRINTER IS 701,80           ! SELECT PRINTER
50 G$=CHR$(27)&"*b256G"       ! GRAPHICS CODES
60 PRINT CHR$(27)&"&19D"      ! SPACING TO 8 DOTS
70 SETGU                       ! SET CRT SCALE TO GRAPHICS UNITS
80 MOVE 0,100                 ! MOVE TO UPPER LEFT
90 FOR L=1 TO 24
100 BREAD A$,32               ! READ 8 ROWS OF DOTS FROM CRT
110 SET I/O 7,16,0           ! ELIMINATE END-OF-LINE SEQUENCE
120 PRINT USING 130 ; G$
130 IMAGE 15X,7A
140 FOR I=1 TO 32
150 A0=NUM(A$[I+0*32])        ! COVERT CHARACTERS TO NUMBERS
160 A1=NUM(A$[I+1*32])
170 A2=NUM(A$[I+2*32])
180 A3=NUM(A$[I+3*32])
190 A4=NUM(A$[I+4*32])
200 A5=NUM(A$[I+5*32])
210 A6=NUM(A$[I+6*32])
220 A7=NUM(A$[I+7*32])
230 FOR J=7 TO 0 STEP -1
240 !                         ! CONVERT HORIZONTAL BYTES TO VERTICAL
250 Q=2^J
260 G=128*(A0\Q MOD 2)+64*(A1\Q MOD 2)+32*(A2\Q MOD 2)+16*(A3\Q MOD 2)
270 G=G+8*(A4\Q MOD 2)+4*(A5\Q MOD 2)+2*(A6\Q MOD 2)+A7\Q MOD 2
280 !
290 IF G=13 THEN G=12        ! CHECK FOR CR CHARACTER AND REPLACE
300 PRINT USING "A";CHR$(G)   ! SEND OUT DATA
310 NEXT J
320 NEXT I
330 SET I/O 7,16,2
340 PRINT USING "K";" "
350 NEXT L
360 SETUU                     ! RESET CRT TO USER UNITS
370 PRINT CHR$(27)&"&16D"      ! SPACING TO 6 LPI
380 END


```

Procedure


1. LOAD and RUN your *application* program (e.g., Standard Pac "HISTO") which generates the CRT graphics you wish to copy.
2. If you have entered data using the application program, make sure it is saved as the following steps will clear the computer memory.
3. Using the LF (linefeed) button on the printer, position the paper to the point where you wish the top of the graph to start and put the printer ON-LINE.
4. LOAD one of the two programs (RCOPY or HCOPY) and press the RUN key.
5. If you have more work to do with your application program you must re-LOAD it.

Below are examples of both the RCOPY and HCOPY output of the program generated from Problem 12.3 in your HP-85 Owner's Manual. The graphs show $Y = \sin(x)/x$.

ROTATED COPY



HORIZONTAL COPY



Ordering Information For Binary Utilities

Binary utilities exist to provide a faster screen copy with a single BASIC command similar to the COPY command on the HP-85. This will allow you to include the command in your application program if it does not already use a binary program. The utilities also provide the capability to read the screen in rows of 8 dots at a time to facilitate partial copying or other special needs. These utilities are available from the appropriate Users' Library listed below under the name "PCOPY". Your dealer or sales representative also has ordering information for Users' Library programs.

When ordering, please fill out all the information below. Include a check or money order for the appropriate amount and send to one of the following addresses:

For Europe, the Mideast and Africa (excluding South Africa):

Users' Program Library Europe
Hewlett-Packard S.A.
7, rue du Bois-du-Lan
P.O. Box CH-1217 Meyrin 2
Geneva, Switzerland

For the U.S. and all others:

Series 80 Users' Library
Hewlett-Packard
1010 N.E. Circle Blvd.
Corvallis, Oregon 97330
U.S.A.

Prices

		(non-members)	(members)
U.S. Users' Library	Disc	\$23.00 U.S.	\$18.00 U.S.
	Tape	\$33.00 U.S.	\$26.00 U.S.
Outside the U.S.	Disc	\$25.30 U.S.	\$19.80 U.S.
	Tape	\$36.30 U.S.	\$28.60 U.S.
Users' Program Library Europe		40 Swiss Francs	

"PCOPY" BINARY UTILITY PROGRAM

PLEASE PRINT:

NAME _____ MEMBER NUMBER _____
COMPANY (if applicable) _____
STREET ADDRESS _____
CITY _____ STATE _____ ZIP _____
COUNTRY _____

Media Type ☐ Disc (5¼)
☐ Tape Cartridge
U.S. Catalog #900-0042

Quantity _____
Amount Enclosed _____
U.P.L.E. Catalog #95087

82905-90022



**HEWLETT
PACKARD**

Personal Computer Division

1010 N.E. Circle Blvd., Corvallis, OR 97330 U.S.A.

Printed in U.S.A.